TI PPC NOTES

NEWSLETTER OF THE TI PERSONAL PROGRAMMABLE CALCULATOR CLUB
P. O. Box 1421, Largo, FL 34649

Volume 14, Number 2 Second Issue 1990/1991

The big news is the availability of the TI-81 which can be ordered from EduCALC but has not yet appeared at retailers in the Tampa Bay area. The TI-81 is a powerful machine with a comprehensive graphics capability, multiple linear regression model solutions without re-entry of data, and matrix algebra which can solve linear equations up to sixth order. You will find a review including benchmark tests on pages 10-12 and sample programs on pages 16 and 26.

This issue also contains listings for those members who agreed to the listings. See pages 7-9, 13 and 28.

Pages 20 and 21 present the results of more investigation by Robert Prins and Scott Garver of the INV ASM problems reported in an earlier issue. All I can say is that you should not be surprised if you have problems when using ASM and INV ASM.

Robert Prins' technique which will generate SBR commands with hexa-decimal addresses appears on page 2. The conversion of a linear regression program for the TI-95 is not complete.

THE TI-81



The TI-74S - V14N1P5 described a new hand-held calculator from TI which is similar to the TI-74 but has a simplified keyboard and the capability to handle ROM modules with larger memory. We stated that the device was called the TI-74B. Tom Ferrio of TI wrote that the device is actually the TI-74S.

AN INADEQUACY IN THE LINEAR EQUATION SOLUTION FROM V8N6P20 - This BASIC routine was extracted from a least squares polynomial solution. It was used successfully in V8N6P18-19 to solve the 7x7 sub-Hilbert as a comparison of the capability of various machines. It was also used without any problems in several least squares solutions for the TI-74 such as the regression with user defined functions on V12N1P14, the interative regression on V12N2P21, the double exponential evaluation of data program on V13N3P19, and the regression with two or three variables on V13N4P22. But, page 18 of this issue reports the inability of the routine to solve linear equations when a diagonal element of the matrix is zero. The problem is caused by the divisions by zero in the solution; for example, at lines 335 and 340 in the listing on page 19. The divides were not a problem when the routine was used in least squares solutions because there the diagonal elements are by definition greater than zero.

So, while the "linear equation" solution routine from V8N6P20 should be useable in all cases when coupled with a least squares solution it will not gnerally be useable as a linear equations solver. If some elements of the input matrix can be zero then the user will be better served by using the linear equation solver from V12N4P13.

GENERATING SBR COMMANDS WITH HEXADECIMAL ADDRESSES - V14N1P24 reported that Robert Prins had found a way to synthesize hex code addresses in user memory. I delayed publication of the details at that time since I had generated the hex code addresses but was unable to get the resulting program to run. With Robert's help I subsequently realized that the running problem was independent of the hex code addresses.

Suppose that we wish to replace the sequence 24637 STO A 'MTH' RUN SBR IND A with the more direct 'MTH' RUN SBR 603D, where either routine provides entry at the NEW option of the INV/LINEAR SYS menu in the MATRIX ALGEBRA portion of the Mathematics Library of the TI-95. First, you must have set the system mode option of the FUNC key. Then enter LEARN mode and proceed to the place you wish to generate the SBR 603D code. Enter the code

```
DFN 1 2 3 4 5 RCB FA60 '='
```

where there is no Fx after the DFN, just the five digits, and the = sign was entered in the ALPHA mode. Use the back arrow key to backstep until the flashing cursor is over the 1. Press 2nd F:CLR F1, and backstep once more. Press 2nd DEL to delete the DFN F1:???@?? leaving the code SBR 603D.

Some insight as to what is going on can be obtained by noting that the Program Code Table on pages C-4 and C-5 of the *TI-95 Programming Guide* shows that FA is the code for the function SBR and 3D is the code for the character =. Now, suppose that you would like to generate the SBR 70A3 command, which is the entry point for the phi subprogram in the Mathematics module. Devise your solution and refer to page 6 of this issue for the answer.

The technique is in direct contradiction of the statement by TI that "... the SBR command can not take a hexadecimal value with a letter value (A-F) in it ..." (see V14N1P5).

3D ON THE fx-7000G - Gene Friel developed this set of programs as an extension of the one on pages 101-103 of the *Graphic Scientific Computers Application Book* (reviewed in V14N1P12). The particular set of four programs at the right plot a torus. Start in Program 0. Enter 25 in response to X÷? and Z÷?, and enter 60 in response to Deg? Don't worry if nothing is plotted immediately. The entire plot takes about nineteen minutes.

Details:

1. The function to be plotted is keyed into Program 1. A is the Z variable. E is the X variable. Positive Z normally goes into the screen depending on the value of angle D. The (M+L)/2, (I+H)/2 and (K+J)/2 terms enable the graph to be displayed vitin other ranges to remove the axes from the screen. Preceding terms are normally A-(M+L)/2 → T, E-(I+H)/2 → U, TcosD + E → X, and TsinD +(K+J)/2 + a function of T, U, and maybe R → Y. Adjust the values of H, I, J, K, L and M in program 0 so that the plot displays properly. For single-valued functions set 0 → R and use R to change the sign of the function for a reflected plot. When T and U are outside the domain of the function, set f(T,U) equal to a large value outside the range values so no point is plotted and no error condition occurs. Using this value enables multiply-connected functions to be plotted.

2. To start the program, execute Program 0. Enter the number of divisions for lines along the X axis for X- and similarly for Z-. Enter the degrees that the Z axis appears above the X-axis for Deg.

3. When the graph is finished the Graph Y=2K at the end of Program 0 keeps the graph in the display. The 2K is just a convenient value above range which keeps the line from appearing on the screen.

4. Change any parameters for a new graph. The Range function will delete the previous graph. Every byte in Program 0 outisde the block between Lbl 2 and Goto 2 at the bottom can be entered by hand to free bytes for a larger function. Since a straight line is drawn between functional values any changes between values aren't shown. Decrease ranges to display rapid local functional changes.

Program 0

```
1→H:8→I:1→J:8→K:
1→L:8→M:1→R:8→N:
"X÷"?→O:"Z÷"?→P:
"Deg"?→D:Deg:(I-
H)÷O→F:(M-L)÷P→G
:Range H,I,2I,J,
K,2K:Lbl 0:" ":D
sz N:Goto 0:Lbl
1:H→E:I→Q:L→Z:M→
B:2→C:Lbl 2:Z→A:
Lbl 3:1→S:Lbl 4:
Prog 2:Prog 1:Pr
og 2:X<H⇒Goto 6:
X>Q⇒Goto 6:Y<J⇒G
oto 6:Y>K⇒Goto 6
:Plot X,Y:S=2⇒Li
ne:S=2⇒Goto 5:Is
z S:Lbl 5:Prog 4
:A≤B⇒Goto 4:Lbl
6:Prog 4:A≤B⇒Got
o 3:E+F→E:E≤Q⇒Go
to 2:F→N:G→F:N→G
:H→Z:I→B:L→E:M→Q
:Dsz C:Goto 2:-R
→R:R=-1⇒Goto 1:G
raph Y=2K
```

Program 1

```
A-(M+L)÷2→T:Tcos
 D+E→X:E-(I+H)÷2
→U:T²+U²→V:V<1⇒G
oto 0:V>9⇒Goto 0
:√(1-(√V-2)²→W:G
oto 1:Lbl 0:99→W
:Lbl 1:Tsin D+(K
+J)÷2+RW→Y
```

Program 2

```
C=1⇒Prog 3
```

Program 3

```
A→N:E→A:N→E
```

Program 4:

```
A+G→A
```

Other 3-D Graphic Functions:

Peak and Hollow:

A-(M+L)÷2 → T:Tcos D+E → X:E-(I+H)÷2 → U:
Tsin D+(K+J)÷2-4Ue-T²e-U² → Y

Try H=0, I=4, J=0, K=4, L=0, M=4, X-=20, Z-=20 and Deg=45.

Attenuated 2-way Cosine

A-(M+L)÷2 → T:Tcos D+E → X: ((E-(I+H)÷2)²+T²) → U:
0-V:U>1⇒9→V:V+Tsin D+(K+J)÷2+e-U(cos 270U)² → Y

Try H=-1; I=1, J=0, K=2.5, L=-1, M=1, X-=25, Z-=25, and Deg=75

CIRCULAR AND NECKLACE PERMUTATIONS ON THE TI-95 PLUS MATH MODULE - Peter Messer

The well-known formula for circular permutations of n different objects is (n-1)!. Of course two circular arrangements are considered equivalent if one can be obtained from the other by simply rotating every object by the same amount and in the same direction. Flipping over (inverting circular arrangements through space is not allowed). Thus, there are (8-1)! = 5040 different ways 8 persons can be seated about a round table.



If instead of people at a table the objects are 8 different colored beads strung on a necklace then there are (n-1)!/2 or 2520 different necklaces. That's because all the (n-1)! non-invertible circular permutations of beads exist as right-handed/left-handed pairs. But if we allow to invert, as in the case of necklaces, then there is no distinction between circular arrangements of opposite handedness. For example, the oppositely handed 2-color arrangements at the right are two distinct circular permutations but they are equivalent necklace permutations.



Circular permutations (non-invertible) and so-called necklace permutations (invertible) are significantly more difficult to solve if not every object is different. For example, if chairs are available in 4 different colors in how many ways can 8 chairs be arranged around a round table? For any arrangement only some colors, or all colors, may occur up to 8 times. A more specific question is "How many circular permutations exist for exactly 1 red, 2 blue, 2 yellow, and 3 green chairs?" Analogously, if we change the "chairs around a table" to "beads strung on a strand" both preceding problems become necklace permutations.

Notations used for program CP1

nTm = non-invertible circular permutations of n objects of m kinds with repetitions allowed. (Example: n = 8 colored chairs, m = 4 colors.)

nIm = invertible necklace permutations of n objects of m kinds with repetitions alowed. (Example: n = 8 beads, m = 4 colors.)

Notations used for program CP2

C = non-invertible circular permutations of $n = x_1 + x_2 + x_3 \ldots$ objects of which x are alike, x are alike, x are alike, etc. (Example: n = 8 chairs, x_1 = 1 red, x_2 = 2 blue, x_3 = 2 yellow, x_4 = 3 green.)

Cs = permutations among C which have bilateral symmetry and thus are not "handed". Examples of bilateral symmetry for two colors are:



N = necklace permutations of $n = x_1 + x_2 + x_3 \ldots$ objects of which x_1 are alike, x_2 are alike, x_3 are alike, etc. For example, substitute "beads" for "chairs" in the above example.

Clearly, $$N = C + \frac{C - Cs}{2} = \frac{C + Cs}{2}$$

Circular and Necklace Permutations - (cont)

User steps for program CP1 (Math cartridge required)

(1) Enter n and m in any order, values remain fixed qand independent of subsequent problems.

(2) View nTm and nIm in any order. Running time is required. If alpha display interferes, press =.

User steps for program CP2 (Math cartridge required)

(1) Enter n, press F1 (n).

(2) Enter x_1, press F2 (x's). Repeat entry of x_2, x_3, ... the same way in any order. The current subscript number of the x-value may be checked by pressing x~t.

(3) User must determine the "greatest common divisor" for all of the x values, enter it, and press F3 (gcd).

(4) Press F4 (EOD) for end of input data. Variable running time follows.

(5) View or print C, Cs, and N in any order by pressing F1 through F3.

(6) Press F5 (NEW) for starting over or for a new problem.

Note: If the user does not follow the exact order of entry steps (1)-(3) incorrect results will occur. The x-values are stored beginning with data register 031.

Sample Problems and Printouts:

CP1

```
n=            8.
m=            4.
nTm=       8230.
nIm=       4435.


n=            8.
m=            4.
nTm=       8230.
nIm=       4435.
m=           12.
nTm=   53750346.
nIm=   26942565.

n=           21.
m=           21.
nTm2.782184 26
nIm1.391092 26
```

CP2

```
n=            8.
x=            1.
x=            2.
x=            2.
x=            3.
gcd=          1.
C=          210.
Cs=           6.
N=          108.

n=           18.
x=            3.
x=            6.
x=            9.
gcd=          3.
C=       226900.
Cs=         280.
N=       113590.

n=           40.
x=            4.
x=            8.
x=           12.
x=           16.
gcd=          4.
C= 2.103278 18
Cs=1745944200.
N= 1.051639 18
```

Circular and Necklace Permutations - (cont)

Program Listings:

Both programs call the phi subprogram in the math module, at step 120 of CP1 and at step 230 of CP2. In the listings that follow the calls of the math module are made using subroutines with hexadecimal addresses generated using the technique developed by Robert Prins and described in more detail elsewhere in this issue. In this case the entry point for the phi subprogram is 70A3. To synthesize that instruction the sequence to be entered in LRN mode, assuming you have already set the system mode, is

DFN 1 2 3 4 5 RCB FA70 LN

Then backstep to the 1, press 2nd F:CLR F1, backstep to the DFN and press 2nd DEL, and see SBR 70A3.

CP1

```
0000 ADV CLR
0002 DFN F1: n @NN
0009 DFN F2: m @MM
0016 DFN F3:nTm@TT
0023 DFN F4:nIm@II HLT
0031 LBL NN STO N `n=`
0038 COL 14 MRG = PRT
0043 RTN
0044 LBL MM STO M `m=`
0051 GTO 0038
0054 LBL II SF 01
0059 LBL TT RCL N SQR
0065 STO E 0 STO S STO F
0072 STO L INC F RCL F
0078 INV IF< E GTO 0137
0084 RCL N / RCL F =
0090 STO G FRC INV IF= L
0096 GTO 0074 RCL F
0101 STO K SBR 0116
0106 RCL G STO K
0110 SBR 0116 GTO 0074
0116 `MTH` RUN SBR 70A3
0123 * RCL M y^x ( RCL N
0130 / RCL K = ST+ S RTN
0137 IF= E GTO 0199
0142 RCL S / RCL N =
0148 STO T TF 01
0152 GTO 0162 `nTm=`
0159 GTO 0038 RCL N /2=
0167 STO J FRC IF= L
0172 GTO 0209 RCL T +
0178 RCL M y^x ( RCL J +
0185 .5=/2= `nIm=` CFG
0196 GTO 0038 RCL E
0201 STO K SBR 0116
0206 GTO 0142 RCL M y^x
0212 RCL J *( RCL M +1=/
0222 4+ RCL T GTO 0188
```

CP2

```
0000 LBL AA ADV CLR 30
0007 STO A DFN F1: n @nn
0016 DFN F2:x's@XX
0023 DFN F3:scd@DD
0030 DFN F4:EOD@ED
0037 DFN F5:NEW@AA `ENT`
0047 `ER` HLT
0050 LBL XX INC A
0055 STO IND A RCL A -30
0063 = x~t RCL IND A `x`
0069 `=` COL 14 MRG =
0074 PRT RTN
0076 LBL nn STO N `n=`
0083 GTO 0070
0086 LBL DD STO J RCL A
0093 STO X RCL J `scd=`
0101 GTO 0070
0104 LBL RR RCL R `C=`
0111 GTO 0070
0114 LBL RS RCL W `Cs=`
0122 GTO 0070
0125 LBL NN RCL I `N=`
0132 GTO 0070
0135 LBL ED RCL J SQR
0141 STO E 0 STO Q STO S
0148 STO F STO L INC F
0154 RCL F INV IF< E
0159 GTO 0259 RCL J /
0165 RCL F = STO G FRC
0171 INV IF= L GTO 0152
0177 RCL F STO K
0181 SBR 0194 RCL G
0186 STO K SBR 0194
0191 GTO 0152 RCL X -30
0199 STO Y = STO Z 1
0205 STO P INC Y
0209 RCL IND Y / RCL K =
0216 x! ST* P DSZ Z
0221 GTO 0207 RCL K `MT`
0228 `H` RUN SBR 70A3 *(
0235 RCL N / RCL K ) x!
0242 / RCL P = ST+ S RTN
0249 RCL E STO K
0253 SBR 0194 GTO 0264
0259 IF= E GTO 0249
0264 RCL S / RCL N =
0270 STO R RCL X -30
0277 STO Y = STO Z 1
0283 STO P 3 STO U INC Y
0290 RCL IND Y /2= STO V
0298 FRC INV IF= L
0302 GTO 0374 RCL V x!
0308 ST* P DSZ Z
0312 GTO 0288 RCL N -
0318 RCL Q =/2= x! /
0326 RCL P = STO W +
0332 RCL R =/2= STO I
0340 DFN CLR
0342 DFN F1: C @RR
0349 DFN F2:Cs @RS
0356 DFN F3: N @NN
0363 DFN F5:NEW@AA `END`
0373 HLT INC Q RCL Q
0378 IF= U GTO 0392
0383 RCL V -.5= GTO 0307
0392 0 GTO 0329
```

MEMBER LISTING - V14N1P11 invited members who were interested to submit their names, addresses, telephone numbers, and interest areas for inclusion in a published membership list. The following listing reflects the response to date:

John Anderson
8750 Walutes Circle
Alexandria VA 22309
Stock market indicators and related programs for the TI-95.

Lee O. Arrington
669 Lawton Road
Hilton NY 14468
(716)-637-4822 H; (716)-477-3147 W
Interpolation, electronics, PC communication on the TI-95.

Cleo Bishop
202 Gilmore Ct. SE
Fort Walton Beach FL 32548
(904)-244-3750
TI-59, -60, -66, -74. HP-41C. IBM PC. Navigation

Herman Blake
26014 S. Dartford Drive
Sun Lakes AZ 85248
Industrial instrumentation, calibration and performance measurement.

Myer Boland
66 Overlook Way
Englishtown NJ 07726
(201)-536-2855
Solution of equations, puzzles, extended precision, interpolation.

William H. Bowen
12610-B Briarglen Loop
Stanton CA 90680-4125
(714)-891-6399 H; (714)-220-2500 W
SR-52, TI-59, electronics applications, data/program exchange with PC's.

Russell N. Byer
411 W. First Street
Marion IN 46952
(317)-664-3121
BASIC programs of all kinds (math, finance, engineering, etc.)

Thomas C. Ceterski
807 West B St.
Iron Mountain MI 49801
(906)-779-9743
TI-58, TI-59. Civil and structural engineering programs.

Victor Chapman
8211 Gulfwood
Houston TX 77075
(713)-991-5363
Curve fitting, chemistry, chemical engineering. TI-59, CC-40, TI-74.

Milton E. Compton
2701 Rigney Road #E-21
Steilacoom WA 98388
(206)-581-3130
Puzzles, alphametics.

Robert C. Crane
28 Greatwood Court, Apt 4
Fairport NY 14450
Financial and investment applications.

Clifford Culpepper, Jr.
1401 Ferncliff Road
Charlotte NC 28211
(704)-366-7350
Compressible fluid mechanics, model aviation, air polution control.

Lucian Dabrowski
1815 North Karlov Ave.
Chicago IL 60639

Willis R. Dortch
4228 East Grant Road, #3
Tucson AZ 85712
(602)-795-8380

Clyde Durbin
3108 Sylvan Ave.
Dallas TX 75212
(214)-748-7837

Membership Listing - (cont)

Jeffrey W. Edelmann
27 Gilfillan St.
Franklin PA 16323
(814)-452-4547
Pharmacy math, pharmacokinetics, astronomy.

Gilbert Farrior
1277 North Warson Road
St. Louis MO 63132
(314)-993-4116
Curve fitting, data analysis, non-parametric statistics.

Norm Foster
Astro Bio at Pier 70
2815 Alaskan Way
Seattle WA 98121
(206)-441-0770
TI-74, great circle navigation, astrology chart programs

Gene F. Friel
3094 Ridge Road
Library PA 15129-9332
(412)-892-4273
TI-59, Casio fx-7000G

William C. Gorman, Jr., President
E-C Apparatus Corp.
3831 Tyrone Blvd. N
St. Petersburg FL 33709
(813)-360-9311 H
Engineering and business programs.

David E. Goss
18004 Fonthill Ave.
Torrance CA 90504
(213)-329-3685
Statistics, numerical methods.

Leo C. Helmer
4333 Itaska Street
St. Louis MO 63116
(314)-351-1432
TI-95. Probability and statistics.

William Hitt
Rt. 1, Box 169-B
Free Union VA 22940
The environment

Morris E. Karp
840 Threadneedle #181
Houston TX 77079-2820
(713)-870-9652
Chemical engineering and instrument engineering programs for the TI-95.

Roger A. Kennedy
329 So. Wisconsin Ave.
Addison IL 60101
(708)-832-0860
Business applications, models, etc.

Linda Kochanski
28476 Cherry
Romulus MI 48174
TI-74, budget/financial programs in BASIC.

Larry Lace
3481 Avenida Ladera
Thousand Oaks CA 91362
(865)-529-4394
Computers, music, skiing, coin collecting.

Hewlett F. Ladd
12 Ridgewood Avenue
Rowayton CT 06853
Celestial navigation; sun and moon ephemerides without nautical almanacs.

Don Laughery
112 W. 2nd Avenue
Baltimore MD 21225
(301)-789-2494 H; (301)-787-3833 W
TI-95 and TI-74, inspection utility, program enhancement for speed.

Laurance M. Leeds (Larry)
10232 El Dorado Drive
Sun City AZ 85351-4243
(602)-933-4379
Unusual problems, multiprecision, programs in BASIC.

A. E. MacKenzie
160 Alto Drive
Oak View CA 93022
(805)-649-3314
Navigation, mathematics, accounting.

Membership Listing - (cont)

Peter Messer
3344 W. Grace Avenue
Mequon WI 53092
(414)-242-1116
Probability and statistics, multi-precision calculations.

Jim Nugent
P. O. Box 118
Peoria IL 61650
(309)-671-8470 W
Tetrahedra, big primes, visualization.

Thomas P. Nyenhuis
Milwaukee Metal Products Co.
8000 West Florist Ave.
Milwaukee WI 53218
(414)-463-2090
TI-95 applications for purchasing and estimating.

Martin Oakes
1335 Wake Forest Way
Fairbanks AK 99709
(907)-479-7732 H; (907)-456-9700 W.
Electronics, surveying, map projections, calendars, modems/communications.

Robert A. H. Prins
Alfred Nobèllaan 112
3731 DX De Bilt
Netherlands
International access code + 31 30 203586
TI-95 assembly language programming.

John Psuik
5018 South 37th Street
Greenfield WI 53221-2531
(414)-281-5779
TI-95, PC interface, PC programming.

J. J. Ramsperger
9361 Candlewood Drive
Huntington Beach CA 92646
(714)-963-8830
TI-95.

Bob Reynolds
6155 Graywood Ave.
Lakewood CA 90712
(213)-866-2104 H; (213)-593-7330 W
RC slope soaring, scale and high performance gliders.

Wayne H. Spyker
RD 2, Box 2152
Stewartstown PA 17363
(717)-993-6514
Solution of machinist equations and model railroading programs.

George Wm. Thomson
5066 Elmhurst Ave.
Royal Oak MI 48073-1102
(313)-435-2070
High accuracy using BASIC dialects.

John Trask
2665 Gardi Street
Duarte CA 91010

Kenneth Wassall
2 Edgewood
Avon CT 06001
(203)-677-1335 H; (203)-693-1665 W
Civil engineering and land surveying applications.

Raymond F. Wenninger (Ray)
111 Oakridge Street
Chicopee MA 01020
(413)-593-6951
Miscellaneous

Bill Wilburn
14333 Domart Ave.
Norwalk CA 90650
(213)-868-0387
Quality control and inspection.

Charlie Williamson
P. O. Box 277177
Sacramenmto CA 95827
(916)-363-2395
Puzzles

Gordon Wilson
Box 967
Christiansted
St. Croix VI 00821-0967
(809)-773-0270
TI-95, statistical sampling, business applications, operations research.

Timothy D. Wolfe
239-A Ft. Mason
San Francisco CA 94123-1308
(415)-928-1226
Curve fitting.

THE TI-81 - Palmer Hanson. The TI-81 is the latest scientific calculator from TI. It is a "graphing calculator" with many of the same capabilities as the Casio fx-7000G, but with much more such as

* Three matrices, each up to 6x6, can be entered and operated on. One result is the ability to solve sixth order simultaneous equations.

* A statistics package with built-in linear, power, exponential and logarithmic regression models, all without reentry of the data when going from one model to another. Up to 150 data points can be entered depending upon how much memory is committed to programs. One caution: the exponential regression model used is y = ab^x not ae^bx as we have used in many of our programs.

* A 2400 byte program capability including labels, tests, branches, gotos, and the like. Up to 37 programs can be stored. The programming language is similar to that of the fx-7000G.

* Up to 27 variables are directly available; A through Z and θ (theta). Five of those (X, Y, R, T and θ are also used in plotting and graphing, so it might seem that there are only 21 variables which are completely under the user's control. Furthermore, these variables cannot be subscripted as with the fx-7000G. However, it turns out that the elements of all three matrices can be used as variables (equivalent to 108 more variables). The {x} and {y} functions (the 2nd 0 and 2nd . keys) can also be used to store and retrieve variables from the statistics input data matrix (up to 150 variables) with one restriction: you cannot store an {x} or {y} with a subscript which is more than one above the highest previously used statistics subscript.

The TI-81 is available from EduCalc for $78.95 plus shipping even though it is not listed in the current catalog (No. 48). It has not yet appeared on the shelves in the Tampa Bay area.

Benchmark Tests:

I tested the TI-81's arithmetic with some of the benchmarks that we have used to test other machines in earlier issues:

1. e x π was equal to π x e indicating that multiplication was commutative. The non-commutative multiply which was unique with the TI-59 was discussed in V8N2P15.

2. sin(45) - cos(45) = -1E-13. It is important to remember that this type of discrepancy can exist when using comparison tests.

3. The square root - squared test: V8N3P13/14 described this test wwhich is a derivative of the $(\sqrt{2})^2$ test by Brian Hayes on page 136 of the January 1981 issue of BYTE. For our test we start with an integer, take the square root five times, take the square five times and compare the result to the original integer. We test selected integers from 2 through 17. The display returned the starting integer in each case. The actual values before truncation to the display were:

2	1.99999 99999 83	12	12.00000 00001 3
3	3.00000 00000 04	13	12.99999 99998 1
5	4.99999 99999 70	15	15.00000 00002 9
7	7.00000 00000 71	17	17.00000 00000 7

where those results are identical with those reported for the TI-95 on V11N3P16 and for the TI-68 on V13N4P10, and all answers are better than those obtained with the TI-59. I performed the calculations with the sequence

$(\sqrt{}\ \sqrt{}\ \sqrt{}\ \sqrt{}\ \sqrt{}\ N)\ x^2\ x^2\ x^2\ x^2\ x^2$

The TI-81 - (cont)

4. 1.0000001 squared 27 times: V9N2P11 described this test from the "Computer Recreations" column of the April 1984 issue of *Scientific American*. The results were:

Exact	674530.47074 10845 59...
Mode A (repeated x)	674530.31804 26
Mode B (repeated ANSxANS)	674530.31804 26
Mode C (N^134217728)	674530.47074 26

The Mode A and Mode B results are the same as with the TI-95, CC-40 and TI-74. The Mode C result is better than with most calculators, and only slightly worse than with the TI-95.

5. The Bob Fruit Benchmark: Bob proposed a compound interest problem as a benchmark in V8N4P4. The appropriate equation is that for the sum of a geometric series S = [(1 + i)^n -1]/i . An annual interest rate of ten percent (i = 0.10/12) and compounding monthly for thirty years (n = 360) yields

Exact	2260.48792 47960 86067 ...
TI-81 using the y^x function	2260.48792 4515

where that answer is slightly better than that from the TI-95, but not as good as that from the TI-59 or TI-68.

6. Speed of calculations:

69!	one second
100p50	one second
328c164	three seconds

where the speed for the combination test is the same as with the TI-68 and a factor of three faster than the TI-95.

7. I tested the capability of the matrix functions with a simultaneous equation solution using a sub-Hilbert matrix (the first row is 1/2, 1/3, 1/4, ..., the second row is 1/3, 1/4, ..., etc.) with ones on the right hand side as suggested by George Thomson in V8N6P18. I entered the terms of the matrix using 1/N - [A](i,j) sequences from the display (see page 6-10 of the guidebook) rather than entering them through the matrix menu (page 6-4 of the guidebook) since the matrix menu will not accept expressions as input. Once a column of ones is entered in the B matrix the solution can be quickly obtained with $[A]^{-1}[B]$ as illustrated on page 9-2 of the guidebook. The previously published results were for the 7 x 7 sub-Hilbert which could be solved by TI-59 programs which called the ML-02 library program as a subroutine and by most other simultaneous equation solutions on the TI-59. Since the TI-81 cannot accommodate a seventh order solution I chose to run the the 5 x 5 sub-Hilbert solutions which was used in the TI-68 evaluation in V13N4P11. The results were

Exact	TI-59	TI-95	TI-81
30	30.00000352680	29.99999464660	29.9999956765
-420	-420.0000381192	-419.9999399311	-419.999951765
1680	1680.000124591	1679.999799175	1679.99983953
-2520	-2520.000158255	-2519.999741040	-2519.99979418
1260	1260.000068682	1259.999886407	1259.999910005
Rel Error	1.18E-7	1.78E-7	1.44E-07

The TI-81 - (cont)

The relative error is calculated as (answer - exact)/exact for each answer and the largest of the five errors entered in the table above. In V9N2P18 James Walters suggested that a better figure of merit might be obtained if the original matrix is multiplied by the solution vector and the result compared with the unity vector. Note that this can be quickly done on the TI-81 by storing the answer from the solution in matrix C and then solving [A][C]. The results for the 5 x 5 sub-Hilbert for the various machines are:

1	1.	0.9999999999	0.9999999640
1	1.0000000013	1.0000000001	0.9999999702
1	0.9999999997	0.9999999998	0.9999999748
1	1.0000000004	1.0000000001	0.9999999781
1	1.0000000005	0.9999999999	0.9999999808
Max error	13E-10	2E-10	360E-10
RMS error	6.6E-10	1.3E-10	271E-10

So, the TI-81 provides coefficients which are approximately as accurate as either the TI-59 or TI-95 for this problem, but the product of the original matrix and the solution vector are two orders of magnitude worse. I do not have an explanation at this time, and don't even know if I am doing something incorrectly. More on this next time. Execution time is another matter. The TI-59 solves the problem in 82 seconds, 57 seconds to find the determinant and 25 seconds to find the solution. The TI-95 needs 16 seconds. The TI-81 solution is available within a second after the sequence $[A]^{-1}[B]$ is initiated.

8. I tested the regression analyses in the statistics section against the sample problem for the TI-95 program in V13N2P16. I received the same results for a, b and r with the exception of the exponential regression where the difference was due to the difference in models.

At this point I feel as though I have only scratched the surface of the capability of the TI-81's 300 functions.

MAILBAG

"I want to commend you on V14N1 of TI PPC NOTES. Great Reading! W.W.

"I recently renewed my membership to your club, but despite a clear note pointing out my new address, you sent the Volume 14, Number 1 issue to my old address. It is only because my mail is temporarily forwarded to my new address that I didn't miss your magazine." G.L.

> I try to keep the addresses straight, but occasionally I make mistakes. Unfortunately, the post office provides forwarding service for shorter periods than in the distant past. Therefore, if you think that you missed an issue or some correspondence please write and ask. I won't be offended.

What a wonderful TI-95 issue. R.P.

PPX PROGRAM 798017 WANTED - This program "Draftsmen's Right Angle Trigonometry (Miter)" was written by William C. Curtis. If you have a copy and would be willing to share it please write to Bill Wilburn, 14333 Domart Ave., Norwalk, CA 90650.

<u>THE</u> <u>598-TEST-1</u> <u>DIAGNOSTIC</u> <u>PROGRAM</u> - V14N1P20 described an anomaly that could occur with this program if the t register was not set to zero prior to starting the program. The original program appeared on page 43 of the *TI 58/59 Service Manual*. We were unable to publish the program at the time due to copyright restrictions. We have now obtained the necessary permission from TI. The program listing follows.

000	76	LBL	026	69	OP	052	43	RCL	078	22	INV	104	10	E'	130	06	6
001	11	A	027	05	05	053	01	01	079	59	INT	105	85	+	131	69	OP
002	00	0	028	01	1	054	72	ST*	080	65	×	106	08	8	132	17	17
003	06	6	029	00	0	055	00	00	081	01	1	107	54	)	133	91	R/S
004	01	1	030	69	OP	056	97	DSZ	082	00	0	108	65	×	134	76	LBL
005	02	2	031	17	17	057	02	02	083	75	-	109	01	1	135	15	E
006	01	1	032	47	CMS	058	12	B	084	07	7	110	00	0	136	85	+
007	01	1	033	03	3	059	76	LBL	085	54	)	111	00	0	137	01	1
008	02	2	034	42	STO	060	13	C	086	77	GE	112	54	)	138	00	0
009	00	0	035	00	00	061	43	RCL	087	15	E	113	44	SUM	139	54	)
010	03	3	036	09	9	062	01	01	088	85	+	114	02	02	140	61	GTO
011	07	7	037	35	1/X	063	94	+/-	089	08	8	115	43	RCL	141	00	00
012	69	OP	038	42	STO	064	74	SM*	090	54	)	116	02	02	142	91	91
013	02	02	039	01	01	065	00	00	091	42	STO	117	69	OP	143	76	LBL
014	01	1	040	09	9	066	73	RC*	092	02	02	118	04	04	144	10	E'
015	07	7	041	06	6	067	00	00	093	43	RCL	119	73	RC*	145	85	+
016	03	3	042	42	STO	068	69	OP	094	00	00	120	00	00	146	01	1
017	06	6	043	02	02	069	10	10	095	65	×	121	69	OP	147	00	0
018	03	3	044	42	STO	070	67	EQ	096	93	.	122	06	06	148	54	)
019	07	7	045	03	03	071	14	D	097	01	1	123	76	LBL	149	61	GTO
020	02	2	046	69	OP	072	43	RCL	098	54	)	124	14	D	150	01	01
021	00	0	047	00	00	073	00	00	099	59	INT	125	69	OP	151	08	08
022	00	0	048	76	LBL	074	65	×	100	75	-	126	30	30	152	00	0
023	02	2	049	12	B	075	93	.	101	07	7	127	97	DSZ	153	00	0
024	69	OP	050	69	OP	076	01	1	102	54	)	128	03	03	154	00	0
025	03	03	051	20	20	077	54	)	103	77	GE	129	13	C	155	00	0

To run the program press A. The title "598-TEST-1 is printed and the diagnostic proceeds. Errors are printed with the incorrect value and the memory location. The program stops with "479.59" in the display. As noted in V14N1P20 you must either set the t register to zero before pressing A or change the Cms at step 032 to a CP to avoid obtaining misleading results.

<u>MORE</u> <u>MEMBER</u> <u>LISTINGS</u>: These listings arrived too late to be assembled with those on on pages 7 through 9.

James E. Coons
1907 Beech St., # 321
Valparaiso IN 46383
(219)-462-6921
Financial Calculations

Richard Bouchard
38 Pecan Drive
Long Beach MS 39560
(601)-868-0128

Thomas W. Dougherty
1007 Hickory
Brigham City UT 84302
(801)-723-8246
Photography, TI-74, Casio fx-502P, Casio fx-4000P.

<u>EDUCALC</u> <u>OFFERS</u> <u>THE</u> <u>fx-7000G</u> <u>APPLICATION</u> <u>BOOK</u> - V14N1P12 reviewed the *Graphic Scientific Coomputer Applications Book* which provides thirty programs for the fx-7000G, fx-7500G and fx-8000G. The price was listed as $5.95, but the only source for the book asked an additional $4.60 for insurance, shipping and handling.

EduCALC now offers the same book (stock number fx-555) for $5.95 plus an additional dollar for surface shipping. You can order with a credit card by calling (800)-633-2252, extension 350, or you can order by mail by writing to EduCALC, 27953 Cabot Road, Laguna Niguel CA 92677.

FLATNESS - P. Hanson. Don Laughery proposed this problem. The idea is to use a best fit plane as a measure of the "flatness" of a surface. The input would be a set of X, Y, and Z points relative to a reference plane. The output would be the deviations from the best fit plane, where the maximum deviations would be of primary interest. Pages 221-222 of Spiegel's *STATISTICS* in the Schaum's Outline Series states:

> "... For example, there may be a relationship between the three variables X, Y and Z which can be described by the equation
>
> $$Z = a_1 + a_2X + a_3Y$$
>
> which is called a linear equation in the variables X, Y and Z.
>
> In a three dimensional rectangular coordinate system this equation represents a plane and the actual sample points (X_1,Y_1,Z_1), (X_2,Y_2,Z_2), ... (X_n, Y_n, Z_n) may "scatter" not too far from this plane which we can call an approximating plane.
>
> By extension of the method of least squares, we can speak of a least squares plane approximating the data. If we are estimating Z from given values of X and Y, this would be called a regression plane of Z on X and Y. ..."

This is, of course, exactly the sort of problem that was solved by the programs in V12N3P19 and V13N3P19. For this special application we made several changes relative to those programs:

* The variable assignment was changed so that X and Y were the independent variables and Z was the dependent variable to agree with user's conventions.

* A routine was added to find the maximum deviations from the best fit plane (steps 605-610).

* A routine was added to permit editing of previously entered data (steps 700-740).

A listing of the program for the TI-74 appears on the opposite page. Two printouts for a sample problem appear at the right. The upper printout is for a problem where one of the values was entered incorrectly. The lower printout is for the same problem with correct input. Note the reduced residuals and standard error.

```
X1 =           1
Y1 =       -9.99
Z1 =        40.16

X2 =           2
Y2 =       -4.98
Z2 =        28.74

X3 =           3
Y3 =           U
Z3 =        17.34

X4 =           4
Y4 =        5.01
Z4 =         5.9

X5 =           5
Y5 =        9.98
Z5 =       -5.53

X6 =           6
Y6 =          15
Z6 =      -16.95

X7 =           7
Y7 =       19.98
Z7 =      -28.37

A1 =  50.8724644
A2 = -11.18359307
A3 = -.0479041414

d1 = -.0074337037
d2 = -.0038408901
d3 =  .0183147992
d4 =  .0019076128
d5 = -.0064157392
d6 = -.0023438842
d7 = -.0001881948

dmax =  .0183147992

dmin = -.0074337037

Mean = -7.142857E-13

S.E. =  .010674434
```

```
X1 =  1
Y1 = -9.99
Z1 =  40.16

X2 =  2
Y2 = -4.98
Z2 =  28.74

X3 =  3
Y3 =  0
Z3 =  17.32

X4 =  4
Y4 =  5.01
Z4 =  5.9

X5 =  5
Y5 =  9.98
Z5 = -5.53

X6 =  6
Y6 =  15
Z6 = -16.95

X7 =  7
Y7 =  19.98
Z7 = -28.37

A1 =  53.55750988
A2 = -12.08011949
A3 =  .1317364727

d1 = -.0013430286
d2 = -.0012232672
d3 =  .0028485882
d4 =  .0029683496
d5 = -.0016424302
d6 = -.0028400336
d7 =  .0012318219

dmax =  .0029683496

dmin = -.0028400336

Mean =  4.285714E-13

S.E. =  .0028507199
```

Interpretations:

The A1, A2 and A3 values in the output are the coefficients of the equation of the plane. A1 is the z-axis intercept, obtained by setting x and y to zero. Similarly, -A1/A2 is the x axis intercept, and -A1/A3 is the y axis intercept.

The residuals from the best fit plane are calculated along the Z axis, not normal to the plane.

Flatness - (cont)

```
10 REM Least Squares Pla
ne with
20 REM Identification of
 the
30 REM Maximum Deviation
s from the Plane
40 REM 10 June 1990
100 DIM A(8,8),B(8),F(8)
,X(50),Y(50),Z(50)
105 INPUT "Use Printer <
Y/N>? ";A$
110 IF A$="Y"OR A$="y"TH
EN PN=1 ELSE 130
115 INPUT "Device Code ?
 ";P$
120 OPEN #1,P$,OUTPUT
130 INPUT "Number of Dat
a Points? ";K
140 FOR I=1 TO K
145 P$=STR$(I)&" = "
150 A$="X"&P$:INPUT A$;X
(I)
160 IF PN<>0 THEN PRINT
#PN,A$,X(I)
170 A$="Y"&P$:INPUT A$;Y
(I)
180 IF PN<>0 THEN PRINT
#PN,A$,Y(I)
181 A$="Z"&P$:INPUT A$;Z
(I)
183 IF PN<>0 THEN PRINT
#PN,A$,Z(I)
185 PRINT #PN:IF E$<>""T
HEN 700
190 NEXT I
200 N=3:PRINT "Solving"
210 FOR I=1 TO N:FOR J=1
 TO N
220 A(I,J)=0:NEXT J
230 B(I)=0:F(I)=0:NEXT I
240 FOR L=1 TO K
250 GOSUB 1000
300 FOR I=1 TO N:FOR J=1
 TO N
305 A(I,J)=A(I,J)+F(I)*F
(J):NEXT J
310 B(I)=B(I)+F(I)*F(0):
NEXT I
315 NEXT L
320 FOR L=1 TO N
325 P=A(L,L)
330 FOR J=L TO N
335 A(L,J)=A(L,J)/P:NEXT
 J
340 B(L)=B(L)/P
345 FOR I=1 TO N
350 IF I=L THEN 375
355 G=A(I,L)
360 FOR J=L TO N
365 A(I,J)=A(I,J)-G*A(L,
J):NEXT J
370 B(I)=B(I)-G*B(L)
375 NEXT I
380 NEXT L
400 FOR I=1 TO N
410 X$="A"&STR$(I)&" = "
420 PRINT #PN,X$;B(I)
430 IF PN=0 THEN PAUSE
440 NEXT I
450 PRINT #PN
500 INPUT "Display Resid
uals <Y/N>? ";A$
510 S1=0:S2=0:MAX=0:MIN=
0
520 FOR L=1 TO K
530 GOSUB 1000
540 ZF=0:FOR J=1 TO N
550 ZF=ZF+B(J)*F(J):NEXT
 J
560 D=F(0)-ZF
570 IF A$="N"OR A$="n"TH
EN 605
580 P$="d"&STR$(L)&" = "
590 PRINT #PN,P$;D
600 IF PN=0 THEN PAUSE
605 IF D>MAX THEN MAX=D
610 IF D<MIN THEN MIN=D
615 S1=S1+D:S2=S2+D*D:NE
XT L
620 PRINT #PN
625 PRINT #PN,"dmax = ";
MAX
630 IF PN=0 THEN PAUSE
635 PRINT #PN
640 PRINT #PN,"dmin = ";
MIN
645 IF PN=0 THEN PAUSE
650 PRINT #PN
655 PRINT #PN,"Mean = ";
S1/K
660 IF PN=0 THEN PAUSE
665 PRINT #PN
670 PRINT #PN,"S.E. = ";
SQR(S2/(K-N))
675 IF PN=0 THEN PAUSE
680 PRINT #PN
700 INPUT "Edit Input Da
ta <Y/N>? ";E$
710 IF E$="N"OR E$="n"TH
EN 800
720 INPUT "Which Data Pa
ir to Edit? ";I
730 IF I<1 OR I>K THEN 7
00
740 GOTO 145
800 INPUT "Delete a Poin
t <Y/N>? ";A$
805 IF A$="N"OR A$="n"TH
EN 900
810 INPUT "Which Point?
";N
815 K=K-1
820 FOR I=N TO K
825 X(I)=X(I+1):Y(I)=Y(I
+1):Z(I)=Z(I+1)
830 NEXT I
835 INPUT "Delete Anothe
r Point <Y/N>? ";A$
840 IF A$="Y"OR A$="y"TH
EN 810
900 INPUT "Print New Inp
ut Table <Y/N>? ";A$
905 IF A$="N"OR A$="n"TH
EN 980
910 FOR I=1 TO K
915 P$=STR$(I)&" = "
920 PRINT #PN:PAUSE ALL
925 A$="X"&P$:PRINT #PN,
A$;X(I)
930 A$="Y"&P$:PRINT #PN,
A$;Y(I)
935 A$="Z"&P$:PRINT #PN,
A$;Z(I)
940 PAUSE 0
945 NEXT I
950 PRINT #PN
980 INPUT "New Solution
<Y/N>? ";A$
990 IF A$="Y"OR A$="y"TH
EN 200
999 STOP
1000 REM LINEAR SOLUTION
 FUNCTIONS
1010 F(0)=Z(L)
1020 F(1)=1
1030 F(2)=X(L)
1040 F(3)=Y(L)
1050 RETURN
```

<u>BEST FIT REGRESSION PROGRAM FOR THE TI-81</u> - This program finds which of the four regression models available provides the "best fit" based on the largest value of the coefficient of determination (r). As such it is the equivalent of the Forecasting - Auto Curve Fit program (RE-11) in the Real Estate/Investment module for the TI-59. It is similar to the five curve fit programs for the TI-74 in V12N4P24-25 and for the TI-95 in V13N2P16-17, but without the ability to fit the hyperbolic (Y = a + b/X), and without the ability to calculate residuals.

The program listing is at the right. The display portions illustrate the ability to use text from the various menus in messages. For example, to assemble the command Disp "r =" which will print the annotation r = the user presses PRGM > 1 to obtain the Disp command, ALPHA " to obtain the leading quotation mark, VARS ▸ ▸ 3 to obtain lower case r, x to obtain the exponent, 2nd TEST 1 to obtain the equal sign, and ALPHA " to obtain the trailing quotation mark.

<u>User Instructions</u>

Enter the data pairs using the Edit option of the statistics menu (2nd STAT), and then run the program to find the best fit.

<u>Sample Problem</u>

The sample problem is the same one used to evaluate the programs in V12N4P24-25 and V13N2P16-17, namely the five points (1, 3.2), (2, 7.4), (3, 12), (4, 16.8) and (5, 22). The output is illustrated at the left below. Note that two lines are required for the output of each parameter together with annotation. This is the same convention used with the fx-7000G. One would like to put the annotation and the corresponding value on the same line, but to date I have not found a method to do that in a program or from the Home Screen. The capability is available somewhere in the TI-81 as illustrated by the output at the right below, which can be obtained by doing an exponential regression from the Home Screen.

```
a=
     3.211745293
b=
     1.196427406
r =
     .9999680549
PwrReg Y=aX^b
```

```
PwrReg
 a=3.211745293
 b=1.196427406
 r=.9999840273
```

Operating from the Home Screen it is also possible to superimpose the regression curve and the input data points on the same plot. Select the regression equation as a function to be plotted by pressing Y= VARS ▸ ▸ 4. Press RANGE and enter the appropriate limits, and press GRAPH to plot the regression equation. To add the input data press 2nd STAT ▸ 2 ENTER. I have been unable to accomplish the same results within a program.

```
Prgm2:BESTFIT
:LinReg
:r²→R
:1→I
:LnReg
:If r²≤R
:Goto A
:r²→R
:2→I
:Lbl A
:ExpReg
:If r²≤R
:Goto B
:r²→R
:3→I
:Lbl B
:PwrReg
:If r²≤R
:Goto C
:r²→R
:4→I
:Lbl C
:If I=1
:LinReg
:If I=2
:LnReg
:If I=3
:ExpReg
:Disp "a="
:Disp a
:Disp "b="
:Disp b
:Disp "r²="
:Disp R
:If I=1
:Disp "LinReg Y=
a+bX"
:If I=2
:Disp "LnReg Y=a
+bLn(X)"
:If I=3
:Disp "ExpReg Y=
ab^X"
:If I=4
:Disp "PwrReg Y=
aX^b"
:End
```

ELLIPSE THROUGH FIVE POINTS - C. Rabe and P. Hanson. Bill Wilburn proposed this problem which comes from a quality control application. Consider the general equation of second degree which can define an ellipse, a parabola, a hyperbola or degenerate forms thereof depending on the magnitudes and signs of the coefficients:

$$(1) \quad Ax^2 + Bxy + Cy^2 + Dx + Ey + F = 0$$

which has six arbitrary constants. Only five are independent since we can divide through by any one of them. You might think that we can arbitrarily select which one; for example, we could divide through by F. That would turn out to be a poor selection; it turns out that the result would be to preclude a solution based on five points which included the origin as one of the input points. In the TI-74 program which was developed we divided through by the coefficient of the x-squared term (A) to yield a modified general equation of second degree

$$(2) \quad x^2 + B'xy + C'y^2 + D'x + E'y + F' = 0$$

where the primed coefficients are simply the unprimed coefficients from equation (1) divided by A. Consulting analytic geometry texts will reveal that

* If B is not equal to zero then the principal axis is not along a coordinate axis.
* B can be eliminated by a coordinate rotation where the rotation angle is $\tan 2\theta = B/(A-C)$.
* If B is equal to zero then the curve is an ellipse if A and C are of the same sign.
* The quantity $(B^2 - 4AC)$ is a constant independent of coordinate rotation.
* If $(B^2 - 4AC) < 0$ then the equation defines an ellipse.
* Those conditions can be applied to equation (2) by recognizing that $A' = 1$.

Analytic geometry texts also typically provide an equation for an ellipse with the major axis parallel to a coordinate axis of the form

$$(3) \quad \frac{(x-h)^2}{a^2} + \frac{(y-k)^2}{b^2} = 1$$

where h and k are the coordinates of the center of the ellipse and a and b are the semiaxes. Again, there are four constants; thus, four points would be required to define such an ellipse. If we multiply through by a^2, expand the two quadratics, and collect coefficients we will have converted equation (3) to a form similar to the modified general equation (2):

$$(4) \quad (1)x^2 + (0)xy + (a^2/b^2)y^2 + (-2h)x + (-2a^2k/b^2)y + (h^2 - a^2 + a^2k^2/b^2) = 0$$

Then, if the appropriate rotation has been performed on equation (2) such that the coefficient of the xy term is zero, then the remaining coefficients of equation (2) can be set equal to the corresponding coefficients of equation (4) and it can be shown that

$$(5) \quad h = -D'/2 \qquad (7) \quad a^2 = (D')^2/4 + (E')^2/(4C') - F'$$

$$(6) \quad k = -E'/(2C') \qquad (8) \quad b^2 = a^2/C'$$

Ellipse through five points - (cont)

Based on the observations above the methodology selected to find the parameters of an ellipse through five points is to

1. Generate a set of linear equations (five equations in five unknowns, B' through F') by substituting the values of each x,y pair into equation (2). The rows of the matrix will be of the form

   $x_i y_i \quad y_i^2 \quad x_i \quad y_i \quad 1$, and the elements of the vector will be $-x_i^2$.

2. Solve for the coefficients of the general equation of second degree.

3. Rotate the coordinates to eliminate the xy term (set B = 0).

4. Convert the coefficients of the reduced equation of second degree to the parameters which define the ellipse.

5. Obtain the coordinates of the center of the ellipse in the original coordinate system by a coordinate rotation.

6. Thus, the five coefficients from the linear equation solution are transformed into the five constants a, b, h, k, and O. The program actually outputs 2a and 2b instead of a and b to meet a user preference.

A BASIC solution was developed for the TI-74 following those guidelines. Carl Rabe also developed a program in TRUE BASIC for a PC which includes a plot of the results. A solution on the TI-95 can be obtained by using the Linear Systems and Conic Sections programs of the Math module. A sample solution from the TI-74 for a problem provided by Bill Wilburn is at the right. The parameters Bill used to generate the five x,y pairs were h = 1, k = 2, 2a = 5.69, 2b = 3.56, O = 15 degrees.

X1 =	.8164
Y1 =	3.7844
X2 =	1.3849
Y2 =	3.861
X3 =	1.9751
Y3 =	3.8571
X4 =	2.5946
Y4 =	3.7433
X5 =	3.2724
Y5 =	3.4122
A =	1
B =	-.7059885942
C =	2.224120745
D =	-.5841439941
E =	-8.199769656
F =	1.1696526
Angle =	14.98669461
maj =	5.690143483
min =	3.555934006
h =	.9987252202
k =	2.001882832
hr =	1.482430831
kr =	1.67552552

A listing of a BASIC program for the TI-74 is on page 19. It turns out that the solution will yield divide-by-zero errors for some cases of zero inputs for either X or Y. The problem is related to an inadequacy in the linear equation solution, in particular the divides at lines 335 and 340. That inadequacy is discussed in more detail on page 2 of this issue. It turns out that for the ellipse through five points problem the difficulty can be circumvented by selecting a sequence of data entry such that the diagonal elements of the matrix can not be zero. Lines 101 through 103 of the program define the rules for the data entry sequence. Lines 155-160 and 175-180 reject data entry which is not consistent with the rules. The program includes two other special features:

* There is no DIM statement even though subscripts are used. Thus, the program relies on the builtin BASIC capability to handle subscripts up to ten.

* The definitions of the angle at line 515 as ATN(1.E+100)/2 and at line 595 as ATN(1.E+100) makes the calculations independent of whether the TI-74 is in the degrees or radians mode. If the user wants to limit the solution to the degree mode he can change those angles to 45 degrees and 90 degrees.

Ellipse through five points - (cont)

```
10 REM Ellipse through 5
 Points
20 REM 7 July 1990
100 REM Data Entry
101 PRINT "Program will
not accept zeroes":PAUSE
 2
102 PRINT "for X1, Y1, Y
2, X3 or Y4.":PAUSE 2
103 PRINT "Adjust order
of data as needed.":PAUS
E 2
105 INPUT "Use Printer <
Y/N>? ";A$
110 IF A$="Y"OR A$="y"TH
EN PN=1 ELSE 140
115 INPUT "Device Code ?
 ";P$
120 OPEN #1,P$,OUTPUT
135 E$=""
140 FOR I=1 TO 5
145 P$=STR$(I)&" = "
150 A$="X"&P$:INPUT A$;X
(I)
155 IF I=1 OR I=3 THEN 1
60 ELSE 165
160 IF X(I)=0 THEN PRINT
 "X"&STR$(I)&" cannot be
 0":PAUSE:GOTO 150
165 IF PN<>0 THEN PRINT
#PN,A$,X(I)
170 A$="Y"&P$:INPUT A$;Y
(I)
175 IF I=3 OR I=5 THEN 1
85 ELSE 180
180 IF Y(I)=0 THEN PRINT
 "Y"&STR$(I)&" cannot be
 0":PAUSE:GOTO 170
185 IF PN<>0 THEN PRINT
#PN,A$,Y(I)
190 PRINT #PN:IF E$<>""T
HEN 850
195 NEXT I
200 REM Set Up Linear Eq
uations
210 PRINT "Solving"
220 FOR I=1 TO 5
230 A(I,1)=X(I)*Y(I)
240 A(I,2)=Y(I)*Y(I)
250 A(I,3)=X(I)
260 A(I,4)=Y(I)
270 A(I,5)=1
280 B(I)=-X(I)*X(I)
290 NEXT I
300 REM Linear Equation
Solution
310 N=5
320 FOR L=1 TO N
325 P=A(L,L)
330 FOR J=L TO N
335 A(L,J)=A(L,J)/P:NEXT
 J
340 B(L)=B(L)/P
345 FOR I=1 TO N
350 IF I=L THEN 375
355 G=A(I,L)
360 FOR J=L TO N
365 A(I,J)=A(I,J)-G*A(L,
J):NEXT J
370 B(I)=B(I)-G*B(L)
375 NEXT I
380 NEXT L
385 B(0)=1
390 D=B(1)*B(1)-4*B(2)
400 REM Print Linear Equ
ation Solution
410 G=65
420 FOR I=0 TO 5
430 X$=CHR$(G+I)&" = "
440 PRINT #PN,X$;B(I)
450 IF PN=0 THEN PAUSE
460 NEXT I
470 PRINT #PN
500 REM Coordinate Rotat
ion
503 IF D<0 THEN 510
505 PRINT #PN,"Not an El
lipse":IF PN=0 THEN PAUS
E
507 PRINT #PN:GOTO 800
510 IF B(1)=0 THEN T=0:G
OTO 525
515 IF B(0)=B(2)THEN T=A
TN(1.E+100)/2:GOTO 525
520 T=(ATN(B(1)/(B(0)-B(
2))))/2
525 CSQ=COS(T)*COS(T)
530 SSQ=SIN(T)*SIN(T)
535 SXC=SIN(T)*COS(T)
540 AP=B(0)*CSQ+B(1)*SXC
+B(2)*SSQ
545 C=(B(0)*SSQ-B(1)*SXC
+B(2)*CSQ)/AP
550 D=(B(3)*COS(T)+B(4)*
SIN(T))/AP
555 E=(-B(3)*SIN(T)+B(4)
*COS(T))/AP
560 F=B(5)/AP
565 HR=-D/2
570 KR=-E/(2*C)
575 AA=D*D/4+E*E/(4*C)-F
580 MAJ=AA:MIN=AA/C
585 H=HR*COS(T)-KR*SIN(T
)
590 K=HR*SIN(T)+KR*COS(T
)
595 IF MIN>MAJ THEN MAJ=
MIN:MIN=AA:T=T-SGN(T)*AT
N(1.E+100)
600 REM Output
605 PRINT #PN,"Angle = "
;T
610 IF PN=0 THEN PAUSE
615 PRINT #PN
620 PRINT #PN,"maj = ";2
*SQR(MAJ)
625 IF PN=0 THEN PAUSE
630 PRINT #PN
635 PRINT #PN,"min = ";2
*SQR(MIN)
640 IF PN=0 THEN PAUSE
645 PRINT #PN
650 PRINT #PN,"h = ";H
655 IF PN=0 THEN PAUSE
660 PRINT #PN
665 PRINT #PN,"k = ";K
670 IF PN=0 THEN PAUSE
675 PRINT #PN
700 REM Print Rotated Ce
nter
710 PRINT #PN,"hr = ";HR
720 IF PN=0 THEN PAUSE
730 PRINT #PN,"kr = ";KR
740 IF PN=0 THEN PAUSE
750 PRINT #PN
800 INPUT "Edit Input Da
ta <Y/N>? ";E$
810 IF E$="Y"OR E$="y"TH
EN 820 ELSE 900
820 INPUT "Which Data Pa
ir to Edit? ";I
830 IF I<1 OR I>5 THEN 8
20
840 GOTO 145
850 INPUT "Edit Another
Point? ";A$
860 IF A$="Y"OR A$="y"TH
EN 820
900 INPUT "Print Input D
ata <Y/N>? ";A$
905 IF A$="N"OR A$="n"TH
EN 955
910 FOR I=1 TO 5
915 P$=STR$(I)&" = "
920 PRINT #PN:PAUSE ALL
925 A$="X"&P$:PRINT #PN,
A$;X(I)
930 A$="Y"&P$:PRINT #PN,
A$;Y(I)
940 PAUSE 0
945 NEXT I
950 PRINT #PN
955 INPUT "Solve with Ed
ited Data <Y/N>? ";A$
960 IF A$="Y"OR A$="y"TH
EN 200
980 INPUT "New Problem <
Y/N>? ";A$
990 IF A$="Y"OR A$="y"TH
EN 135
999 END
```

MORE ON PROBLEMS WITH THE ASM / INV ASM FUNCTIONS - Scott Garver and Robert Prins

Editor's Note: V13N1P8 reported an instance in which the INV ASM function would produce non-existent labels. The following excerpts from letters between members document the process by which the source of the problem has been identified.

Robert asks if rewriting the INV ASM program for use in his Utilities Cartridge is worth the amount of memory required. My inclination is to say no. I just haven't used the ASM capability, probably because the TI-95 is so fast with respect to the TI-59 that I just didn't see the need. Do other members use the ASM function? Has any member performed any benchmark tests to establish the increase in execution speed which is attained after a program is modified with the ASM function?

Letter, Scott Garver to Palmer Hanson and Robert Prins - 24 February 1990

I HAVE COME ACROSS A RATHER INTERESTED AND REPEATABLE ERROR THAT IS GENERATED WHILE USING THE <INV ASM> FUNCTION. WHILE DOCUMENTED SOME TRANSLATED PROGRAMS, I <ASM>ed A PROGRAM TO CALCULATE THE SPEED-UP RATIO. THE ORIGINAL PROGRAM SEGMENT WAS (WITH PARTITIONING OF P0416,R660,F1504 or P4096,R200,F1504):

... SBL *4 DFN F5:in_@_B RTN LBL _B STO J ...
with STO J at location 0321.

AFTER PRESSING <ASM> THE SEGMENT IS:

... SBL *4 DFA F5:in_@0321 RTN LBL _B STO J ...
which is correct.

HOWEVER, <INV ASM> PRODUCES:

... SBL *4 DFN F5:in_@_E RTN LBL _B STO J ...
where the DFN statement is wrong and
points to a non-existent label name.

AND PRESSING <ASM> NOW DOES NOT ALTER THIS DFN STATEMENT. WHEN I SHIFT THE WHOLE PROGRAM BY INSERTING A <NOP> AT 0000, THIS ERROR DOES NOT OCCUR. A SIMPLE, REPEATABLE, EXAMPLE OF THIS ERROR IS OBTAINABLE BY ENTERING AT 0310 {... **DFN F5:XXX@AS RTN LBL AS ...**} THEN PRESS **<ASM><INV ASM>** THIS YIELDS {... **DFA F5:XXX@0321 RTN LBL AS ...**}. (THIS ASSUMES A PARTITIONING WITH F1504.) THUS SHOWING THAT **<INV ASM>** FAILS TO WORK IN THIS CASE. IF YOU DO THE SAME STEPS, BUT AT LOCATION 0311, IT WORKS (NO ERROR). ALSO, IF THE PARTITIONING ALTERS THE LOCATION OF THE <PGM> WITHIN THE RAM STACK, THE ERROR DOES NOT OCCUR. IF YOU SHIFT THE PARTITIONING TO P4096,R199,F1512 (INCREASE FILE SIZE BY 8), THIS SHOULD CAUSE THE SAME ERROR 8 STEPS LOWER (0310 to 0302). AND IT DOES!! RETURNING THE PARTITION TO P4096,R200,F1504 AND STARTING THE CODE AT 0310, AS BEFORE, THE ERROR REOCCURS.

IN TRYING TO UNDERSTAND THIS I NOTICED THAT THE LAST BITE OF THE LBL AS IS AT h2B00. HEXCODE AND LOCATION IS:

```
2AF6:  F7  94  58  58  58  41  53   BC   FF  41  53
       DFN F5  X   X   X   A   S   RTN  LBL  A   S
```

IN FACT, FOR A LABEL AT LOCATION h2AFE-h2B00 THE CALCULATOR CANNOT CORRECTLY OR AT ALL **<INV ASM>** FROM IT IRREGARDLESS OF WHERE THE **DFN**, **SBL**, or **GTL** IS LOCATED. A HIDDEN BENEFIT I FOUND IS TO USE R. PRINS' UTILITY <CRC> FUNCTION TO DETERMINE IF THIS ERROR IS OCCURRING BY PRESSING <CRC><ASM><INV ASM><CRC>. IF THE CHECKSUMS DON'T MATCH THE ERROR HAS OCCURRED.

More on Problems with the ASM / INV ASM Functions - (cont)

I RECALLED A WRITE-UP OF A SIMILAR ERROR IN THE NOTES. I FOUND ON V13N1P8 WHERE THE SAME ERROR IS DISCUSSED. HOWEVER, ON V13N1P8 THE ERROR OCCURS IS AT h2FFE-h3000. THE ERROR I ENCOUNTERED OCCURS WHEN THE LABEL IS AT h2AFE-h2B00. THIS CAUSED ME TO WONDER HOW MANY OTHER LOCATIONS THIS ERROR WOULD OCCUR. THE DIFFERENCE BETWEEN THESE LOCATIONS IS H500. SO, I WOULD EXPECT THE SAME AT LOCATIONS h2FFE +- (h500*n) WHERE n = 1, 2, 3,...

WITH PARTITIONING P7200,R000,F0000, THIS GIVES US LOCATIONS:

MEMORY MAP	PC LOC'N
h25FE	0542
h2AFE	1822
h2FFE	3102
h34FE	4382
h39FE	5662
h3EFE	6942

TESTING THESE LOCATIONS WE FIND OUT THAT THEY ARE ALL ERROR LOCATIONS.

THIS CAUSED ME TO SPECULATE ON EVEN MORE LOCATIONS. I GUESSED THAT THE hXXFE LOCATIONS IS THE MOST GENERAL SET. (I AM GUESSING THAT THE ERROR IS CAUSED BY THE LAST BITE OF THE LABEL BEING AT hXX00, THIS BEING CONSISTENT WITH MR. POPOV'S AND MY PROBLEM.) TESTING LOCATIONS h26FE (pc0798) AND h23FE (pc0030) FOUND MY SPECULATION TO BE CORRECT.

SO THE NEW GENERAL EQUATION OF THE LOCATIONS OF LABELS THAT CANNOT BE <INV ASM> IS:

h23FE + (h100 * n) where n = 0, 1, 2,..., 27 dec (1B hex)

THUS WE SEE THERE ARE 28 LOCATIONS FOR A LABEL TO RESIDE WHERE THIS ERROR WOULD OCCUR.

Letter, Robert Prins to Scott Garver, copy to Palmer Hanson, 2 April 1990

Let's start with your observations on the bug in the INV ASM function, with help of your scrutinizing I have been able to locate the bug in the coding of the INV ASM function. The actual buggy code is below:

```
DBA8 D3 INC   R119
     77
DBAA 79 ADC   %>00,R119    <== This should have been R118
     00
     77
```

As a result of the above, the TI-95 will, if the address of the first character of the label is of the type xxFFh, look 255 bytes back for the second character. If the code at this location happens to be in the range 20h..7fh, it will be inserted into the GTL, SBL or DFN! If it falls outside this range, the absolute address will remain an absolute address!

Now that we know the source of the bug, there is an easy cure, I could rewrite INV ASM (I have already done so) and include it in the UC, but my question is: "Would the approximately 240 bytes required to do so be useful?"

REDUCTION IN PPX PROGRAM AVAILABILITY - Maurice Swinnen writes that he sent all of his TI-59 material to a school in Belgium. Thus the programs listed in V13N1P10-20 as available from him (Code 2) are no longer available. There were about fifty programs for which he was the only source.

10 POINT GAUSS QUADRATURE INTEGRATION FOR THE TI-59 - R. Prins

This program provides a numerical integration capability for the TI-59 including an option to calculate inproper integrals from a lower boundary to infinity. The program and the ten constants at the right are entered with partitioning 7 Op 17 and recorded with partitioning 6 Op 17 in bank 2. Note the constants are to be entered to thirteen digit accuracy; for example, the first constant in the table can be synthesized in the display register as 0.1488743389 + 816E-13 = before storing to data register 60. The INS commands (code 46) at steps 308 and 366 can be synthesized with the sequence RCL 46 BST BST 2nd Del SST.

R60	0.1488743389816
R61	0.2955242247148
R62	0.4333953941292
R63	0.2692667193100
R64	0.6794095682990
R65	0.2190863625160
R66	0.8650633666890
R67	0.1494513491506
R68	0.9739065285172
R69	0.0666713443087

User Instructions

1. The function to be integrated must begin with LBL A' and end with RTN. When the function is called the value of x is only in the display. To use it at other points in the integration function you can store it in R00 or R10 through R59. Both = and CLR can be used in the function. The LBL A' can be located anywhere in bank 1, but should be located at step 000 for maximum speed.

2. Read the program and the constants into bank 2.

3. Enter the lower limit of integration and press A.

4. Enter the upper limit of integration and press B. Note that steps 3 and 4 must be performed in the order indicated. If integrating an improper integral is required, step 4 must be omitted.

5. Enter the number of intervals and press C.

6. Press D to integrate. The program stops with the integral in the display.

Program Listing

```
000  76 LBL
001  16 A'
002  70 RAD
003  42 STO
004  00  00
005  53  (
006  01  1
007  85  +
008  93  .
009  05  5
010  65  ×
011  53  (
012  01  1
013  00  0
014  65  ×
015  89  π
016  65  ×
017  43 RCL
018  00  00
019  54  )
020  38 SIN
021  54  )
022  35 1/X
023  92 RTN
```

```
240  76 LBL      272  17  17       304  06  06       336  49 PRD       368  43 RCL
241  14  D       273  00  0        305  54  )        337  09  09       369  02  02
242  87 IFF      274  42 STO       306  16 A'        338  06  6        370  33 X²
243  01  01      275  09  09       307  53  (        339  69 OP        371  61 GTO
244  02  02      276  43 RCL       308  46 INS       340  17  17       372  03  03
245  70  70      277  03  03       309  65  ×        341  43 RCL       373  09  09
246  53  (       278  44 SUM       310  69 OP        342  09  09       374  76 LBL
247  43 RCL      279  06  06       311  25  25       343  92 RTN       375  11  A
248  01  01      280  22 INV       312  73 RC*       344  53  (        376  42 STO
249  75  -       281  86 STF       313  05  05       345  87 IFF       377  01  01
250  53  (       282  00  00       314  54  )        346  00  00       378  02  2
251  53  (       283  05  5        315  44 SUM       347  03  03       379  42 STO
252  94 +/-      284  09  9        316  09  09       348  50  50       380  07  07
253  85  +       285  42 STO       317  97 DSZ       349  94 +/-       381  86 STF
254  43 RCL      286  05  05       318  08  08       350  85  +        382  01  01
255  02  02      287  05  5        319  02  02       351  01  1        383  43 RCL
256  54  )       288  42 STO       320  90  90       352  54  )        384  01  01
257  55  ÷       289  08  08       321  01  1        353  35 1/X       385  92 RTN
258  43 RCL      290  69 OP        322  94 +/-       354  42 STO       386  76 LBL
259  04  04      291  25  25       323  49 PRD       355  02  02       387  12  B
260  55  ÷       292  53  (        324  07  07       356  65  ×        388  22 INV
261  42 STO      293  73 RC*       325  22 INV       357  02  2        389  86 STF
262  03  03      294  05  05       326  87 IFF       358  85  +        390  01  01
263  02  2       295  87 IFF       327  00  00       359  43 RCL       391  42 STO
264  54  )       296  01  01       328  02  02       360  01  01       392  02  02
265  42 STO      297  03  03       329  81  81       361  75  -        393  92 RTN
266  07  07      298  44  44       330  97 DSZ       362  01  1        394  76 LBL
267  54  )       299  65  ×        331  04  04       363  54  )        395  13  C
268  42 STO      300  43 RCL       332  02  02       364  16 A'        396  42 STO
269  06  06      301  07  07       333  76  76       365  53  (        397  04  04
270  07  7       302  85  +        334  43 RCL       366  46 INS       398  92 RTN
271  69 OP       303  43 RCL       335  07  07       367  65  ×        399  00  0
```

10 Point Gauss Quadrature Integration for the TI-59 - (cont)

Sample Problem

The function 1/(1 + .5sin(10πx) integrated from 0 to 1 was used as a sample problem. The integral is exactly 2/ 3 = 1.154700538379... . The program can find the integral correct to ten digits in 7.5 minutes using 15 intervals.

The 7 point Gauss quadrature integration program in the Mathematics module for the TI-95 can find the integral of the same function correct to ten digits in 84 seconds using 17 intervals.

FUNCTION ANALYSIS PROGRAM - R. Skarda. The discussion of the Gauss quadrature programs for the TI-59 and TI-95 provides background for a very large (2352 steps) function analysis program for the TI-95 submitted by Richard Skarda. For an input function the program provides:

* A plotting capability.

* Calculation of first and second derivatives at a selected point.

* Trapezoidal rule integration.

* Simpson's rule integration.

* Romberg integration.

* Gauss Legendre quadrature integration with options to use n = 2, 4, 8, 16, 32, 64, 128, 256 or 512 .

* A routine to calculate the data base for the Gauss Legendre integration. The routine runs for over 37 hours and essentially fills an 8K module.

Editor's Note: Early tests show that the Gauss-Legendre quadrature option provides faster solutions than the Gauss quadrature program in the TI-95 Math module, even when using the n=2 option. I plan to publish the program in the next issue. In the meantime, if you would like to experiment with the program send two dollars and I will send a magnetic tape of the program. Sorry, I do not yet have a tape or a module with the Gauss Legendre constants.

A NEW SOURCE FOR TI-95 HARDWARE - Dan Eicher wrote: "TI has turned its entire inventory of TI=95's over to an outfit called Starlight of America who is liquidating them. ..." I called 1-800-TI-CARES for confirmation and was given the number 1-800-869-7343 for Starlight of America. They are selling TI-95's for $59.95 and the cartridges (mathematics, statistics and chemical engineering) for two for $14.95 or three for $21.95. There are no shipping charges and you may order cartridges without ordering calculators. They promise delivery in about two weeks after your check has cleared. Their address is:

Starlight of America
11131 Dora Street
Sun Valley CA 01352

I suggest that you call to check availability before ordering. Please mention our club if you do order.

LOOKING BACK - P. Hanson. It was the purchase of an older hand-held calculator at an auction which triggered this trip down memory lane. The SR-10 came complete with a switchable 115V/230V charger and a manual with an entry which showed an original purchase date of 8/6/73. The device requires plug-in NiCad AA cells (no battery packs, praise the Lord!) when used with the charger, but works perfectly well with standard AA cells. A battery saver circuit is provided which turns off the LED display about thirty seconds after the last entry (15-60 seconds, per the manual). The four arithmetic functions are supplemented by 1/x, x^2, and $\sqrt{x}$. The manual includes expressions for calculation of trigonometric, logarithmic and exponential functions. There are no parentheses. The calculator uses "adding machine logic" (see V9N1P4 and V9N3P17), not A.O.S. The seventeen year old device is in amazingly good condition with every key responding firmly and with an audible click.

While working with the device I started thinking about the so-called "good, old days" although it is always one of my contentions that the old days weren't so good. I remembered writing my first digital computer program as a graduate student at the University of Minnesota in early 1960. The computer was a RemRand 1103 which used tubes for the logic, CRT's for fast access memory, a magnetic drum for storage of data and programs, and paper tape devices for input and output. The computer filled a sizeable building. Students never touched the computer. We wrote our program, punched a paper tape for entry, and respectfully submitted the tape to the operator across a counter. We were required to estimate the run time as well, and if the paper tape output did not begin to appear when anticipated we might have our run shut down. Our class of some thirty students worked in teams of three students -- it was an imposition enough to tie up the machine for ten student programs -- thirty student programs would have been entirely out of the question. Course grades were largely a function of the success of the programs. If the program ran successfully on the first submittal then A's for the team members were a virtual certainty. If not, then the highest grade that the team members could receive was a B, etc.

The following summer I returned to work with Honeywell and was assigned to the St. Petersburg facility. My first system assignment was on an inertial navigation system for an Army drone. The digital computer used magnetic drum memory, was programmed in machine language, and input/output was in hexadecimal. Work with the system provided my first exposure to real time programming.

For several years Ralph Snyder was recognized as the senior member of our club. But in 1985 he did not renew his membership citing his age (85 at that time) and limited interests. V10N3 identified Larry Leeds at 80 as the new senior member . Larry remains as active as ever at 85. I thought it might be interesting to go beyond the age category and identify other measures of long term association with our club interests. So, I am suggesting that members send in their claims to the following categories:

(1) Oldest member, if you can beat Larry's record of 85.

(2) Youngest member.

(3) Longest membership - identify the first issue of 52 *Notes* received by normal distribution; do not include back issues received.

(4) First digital computer program - type of program, computer, date, etc.

(5) First hand-held programmable calculator program - type, machine, date, etc.

I will report the "winners" in a coming issue.

ADDENDUMS FOR THE TI-74 MATH MODULE - Ron Burnham writes that he found three addendums with his TI-74 Mathematics Library Guidebook. I only received one which had the notation 1059977-0101 in the lower right hand corner. Copies of the two which I did not have are reproduced below.

Addendum: TI-74 Math Library Cartridge (Continued)

Gamma Function
See pg. 2-8

The Gamma Function Program (GAM) can calculate answers for negative non-integers. The magnitude of the answer is correct, but the sign of the answer is not always correct. To avoid calculating an incorrect answer for gamma, use the following program instead of the program built into the cartridge. This program calls a subprogram that is in the Math Library Cartridge.

```
200 RAD: INPUT "Enter X: ";X
210 CALL GAMS(X,G)
220 IF G>294.73 THEN 230 ELSE 240
230 DISPLAY "Ln(Gamma(X)) = ";G:PAUSE:GOTO 200
240 GM = SGN(X)^INT(X)*EXP(G)
250 DISPLAY "Gamma(X) = ";GM:PAUSE:GOTO 200
```

To calculate gamma:

1. Enter the program.
2. With the Math Library cartridge installed, run the program.
3. Enter X as prompted.
4. View the result.
5. If you have more gamma calculations to perform, press [ENTER] and return to step 3.
6. If you are finished with gamma calculations, press [BREAK] to stop the program.

Runge Kutta

The Runge-Kutta Program requires that you enter a step size that causes no long or short step to occur at the end of the interval. Otherwise, the program calculates the last step incorrectly. To determine a step size that causes all steps to be equal in width, choose an integer number of steps to be between Xmin and Xmax and use that number in the following equation.

Step size = (Xmax − Xmin)/number of steps

Addendum 2: TI-74 Math Library Cartridge

Complex System Program

The COMPS program may return the message MATRIX IS SINGULAR even though there is a solution for a specific matrix.

If this occurs, you can solve your complex system using option 5 of the Matrices (MAT) program. Using the method shown below, manually expand your $n \times n$ system of complex equations into a $2n \times 2n$ system of real equations.

Each complex coefficient is evaluated as

$$(a_{11}, b_{11}) = \begin{matrix} a_{11} & -b_{11} \\ b_{11} & a_{11} \end{matrix}$$

Using this method, the system

$$\begin{matrix} (a_{11}, b_{11}) & (a_{12}, b_{12}) & (a_{13}, b_{13}) & & (c_1, d_1) \\ (a_{21}, b_{21}) & (a_{22}, b_{22}) & (a_{23}, b_{23}) & & (c_2, d_2) \\ (a_{31}, b_{31}) & (a_{32}, b_{32}) & (a_{33}, b_{33}) & & (c_3, d_3) \end{matrix}$$

is expanded to

$$\begin{matrix} a_{11} & -b_{11} & a_{12} & -b_{12} & a_{13} & -b_{13} \\ b_{11} & a_{11} & b_{12} & a_{12} & b_{13} & a_{13} \\ a_{21} & -b_{21} & a_{22} & -b_{22} & a_{23} & -b_{23} \\ b_{21} & a_{21} & b_{22} & a_{22} & b_{23} & a_{23} \\ a_{31} & -b_{31} & a_{32} & -b_{32} & a_{33} & -b_{33} \\ b_{31} & a_{31} & b_{32} & a_{32} & b_{33} & a_{33} \end{matrix} \times \begin{matrix} X_1 \\ Y_1 \\ X_2 \\ Y_2 \\ X_3 \\ Y_3 \end{matrix} = \begin{matrix} c_1 \\ d_1 \\ c_2 \\ d_2 \\ c_3 \\ d_3 \end{matrix}$$

Load the values from the manual expansion of your system into the Matrices (MAT) program, option 5, to solve the matrix. The results of the MAT program should be interpreted as follows:

- Odd-numbered results (X_1, X_3, X_5, etc.) are the real parts (X_1, X_2, X_3, etc.).
- Even-numbered results (X_2, X_4, X_6, etc.) are the corresponding imaginary parts (Y_1, Y_2, Y_3, etc.).

ELLIPSE THROUGH 5 POINTS ON THE TI-81 - P. Hanson

This program is the equivalent of the TI-74 program on pages 17 through 19 of this issue, but with an added capability to plot the ellipse. The program requires that the five data points have been entered as statistical data. The 733 byte program is listed at the right hand side, with line numbers (actually command numbers) added to facilitate discussion of the program. In the listing I used the same convention as in the fx-7000G program on page 3 and the TI-81 program on page 16, namely, I made the listing look as much as possible like that in the display as an aid to checking the entry of the program into the calculator.

```
   Prgm1:ELLIPSE
   :IF Dim{x}=5
   :Goto A
   :Disp "n≠5"
   :Pause
 5 :Lbl A
   :5→Arow
   :5→Acol
   :5→Brow
   :1→Bcol
10 :5→Crow
   :1→Ccol
   :1→I
   :Lbl B
   :{x}(I)*{y}(I)→[
   A](I,1)
15 :{y}(I)*{y}(I)→[
   A](I,2)
   :{x}(I)→[A](I,3)
   :{y}(I)→[A](I,4)
   :1→[A](I,5)
   :-{x}(I)*{x}(I)→
   [B](I,1)
20 :IS>(I,5)
   :Goto B
   :[A]⁻¹[B]→[C]
   :If [C](1,1)²-4*
   [C](2,1)<0
   :Goto C
25 :Disp "NOT AN EL
   LIPSE"
   :Pause
   :Lbl C
   :Deg
   :If [C](1,1)≠0
30 :Goto D
   :0→θ
   :Goto F
   :Lbl D
   :If [C](2,1)≠1
35 :Goto E
   :45→θ
   :Goto F
   :Lbl E
   :tan⁻¹([C](1,1)/
   (1-[C](2,1)))/2→
   θ
40 :Lbl F
   :(cos θ)²+[C](1,
   1)sin θcos θ+[C]
   (2,1)*(sin θ)²→A
   :((sin θ)²-[C](1
   ,1)sin θcos θ+[C
   ](2,1)*(cos θ)²)
   /A→C
   :([C](3,1)*cos θ
   +[C](4,1)*sin θ)
   /A→D
   :(-[C](3,1)*sin
   θ+[C](4,1)*cos θ
   )/A→E
```

User Instructions

1. Clear the statistical data (ClrStat) and then enter the five data points through the (Edit) option of the STAT DATA menu.

2. Run the program. The calculator will stop with the Angle, the major axis and the minor axis on the Home Screen. Press ENTER and the calculator will display the center of the ellipse in the original coordinate system (the H and K values). Press ENTER again and the calculator will display the center of the ellipse in the rotated coordinate system used in the solution (the Hr and Kr values). Press ENTER one more time and see the flashing cursor.

3. Press GRAPH and see the ellipse plotted on the Home Screen. You may now operate on the plot with the ZOOM and TRACE functions.

Comments on the program

Line 1 checks Dim{x} to see that five data points have been entered into statistical data. Actually, this is not exactly so as suggested in the note at the bottom of page 7-11 of the *TI-81 Graphics Calculator Guidebook*. The value in Dim{x} is incremented when an x value is entered, and is not changed when the corresponding y value is entered. Therefore, if you entered the fifth x value but failed to enter the fifth y value for some reason the value in Dim{x} would be 5 and the program would proceed with the value of 1 provided for {y}(5) by the statistics routine.

Lines 6 through 11 set the dimensions of the matrices for the solution of the five simultaneous equations.

Lines 13 through 21 use the input x and y values from the statistical data to generate the matrix A and the vector B which define the set of linear equations. The user needs to be careful to recognize the differences between the use of parentheses () to indicate subscripts, brackets [] to indicate matrices, and braces {} to indicate statistical data.

Ellipse through 5 Points on the TI-81 - (cont)

Line 22 solves the set of linear equations using the technique defined on page 9-2 of the guidebook.

Line 23 tests the value of the determinant ($B^2 - 4AC$) to find whether or not the solution is for an ellipse. Note that A=1 as explained on page 17.

Lines 28 through 39 calculate the angle for a coordinate rotation which will reduce the coefficeint of the xy term to zero.

Lines 40 through 45 perform the coordinate rotation. I inserted some multiplication signs which are not necessary but which I thought added clarity to the listing.

Lines 46 through 49 find the center of the ellipse in the rotated coordinate system (variables R and S), and then perform the coordinate transformation to calculate the center of the ellipse in the original cordinate system (variables H and K).

Lines 50 and 51 calculate the squares of the semiaxes of the ellipse (variables M and N) but do not define which is the major axis.

Lines 52 through 62 identify the major and minor axes and place the major axis in M and the minor axis in N. The identification of which axis is which is performed using the squares of the semiaxes. The angle is adjusted to be the angle to the major axis.

Lines 56 through 58 calculate the sign of the angle since the TI-81 does not have a signum function.

Lines 69 through 82 display the results in the Home Screen. The lower case r's in the annotations "Hr=" and "Kr=" are obtained from the VARS menu, and the ='s in those annotations are obtained from the TEST menu. Assuming the user in in the programming mode the key sequence to generate "Hr=" is ALPHA " ALPHA H VARS ▶ ▶ 3 2nd TEST 1 ALPHA " .

Lines 83 through 99 set up the TI-81 to plot the ellipse relative to the original coordinate system.

Lines 84 and 85 prevent any existing second and third parametric equation sets from being plotted.

Lines 86 and 87 calculate the semiaxes.

Lines 88 through 96 set up the range for the plot. The 1.5 multiplier in the Xmax and Xmin calculations ensure that the scales for the horizontal and vertical axes are the same.

Lines 97 and 98 place the parametric equations into the Y= list.

```
45  :[C](5,1)/A→F
    :-D/2→R
    :-E/(2C)→S
    :Rcos θ-Ssin θ→H
    :Rsin θ+Scos θ→K
50  :D²/4+E²/(4C)-F→
    M
    :M/C→N
    :If M>N
    :Goto G
    :M→N
55  :N/C→M
    :-1→Z
    :If θ>0
    :1→Z
    :θ-90Z→θ
60  :Lbl G
    :2√M→M
    :2√N→N
    :ClrHome
    :Disp "ANGLE="
65  :Disp θ
    :Disp "MAJOR AXI
    S="
    :Disp M
    :Disp "MINOR AXI
    S="
    :Disp N
70  :Pause
    :ClrHome
    :Disp "H="
    :Disp H
    :Disp "K="
75  :Disp K
    :Pause
    :ClrHome
    :Disp "Hr="
    :Disp R
80  :Disp "Kr="
    :Disp S
    :Pause
    :Param
    :X2T-Off
85  :X3T-Off
    :M/2→A
    :N/2→B
    :0→Tmin
    :360→Tmax
90  :5→Tstep
    :1.5(H-A-1)→Xmin
    :1.5(H+A+1)→Xmax
    :1→Xscl
    :K-A-1→Ymin
95  :K+A+1→Ymax
    :1→Yscl
    :"Acos Tcos θ-Bs
    in Tsin θ+H"→X1t
    :"Acos Tsin θ+Bs
    in Tcos θ+K"→Y1t
    :ClrDraw
100 :End
```

THE SeTIc DOCUMENTATION FOR THE TI-95 MATH MODULE - V12N4P3 reported that SeTIC, a successor club to TISOFT, had published a program listing and flow chart for the Mathematics module of the TI-95. The original printing has been sold and SeTIc is no longer operating. Through the help of Thomas Coppens and Robert Prins our club has obtained nine new copies of the documentation. The price is eight dollars.

PRINTOUTS OF CARTRIDGE CONTENTS - John Psuik has generated printouts of all three cartridges as well as the TI-95 RAM/ROM. He also has written a software package which will convert test files into hex code. If you want copies write to him for details and costs at 5018 South 37th Street, Greenfield WI 53221-2531 or call him at (414)-281-5779 after 5 PM Central Time.

PRINTED MATERIAL FOR SALE - The club has received several donations of printed material. Typically this material is TI-59 related where the donor is no longer using that device. Most of the material has been used to some extent. Other members can obtain this material by paying for estimated shipping costs. The available material includes:

* A copy of the so-called "Fish Book". Send three dollars.

* A set of the issues from *PPX Exchange* covering the period from Volume 5 Number 1 through Volume 6 Number 6. These have some user comments added but otherwise are like new. Send three dollars.

* A paperback copy of TI's *Sourcebook for Programmable Calculators*. This unit is shopworn but useable. Send three dollars.

ONE MORE MEMBER LISTING - Perry K. Gerhart, RD #2, Box 497, Breinigsville PA 18031. Telephone (215)-285-6736. Land surveying programs for the TI-59 and TI-95

HP PAPER FOR THE PC-324 - V14N1P13 discussed Hewlett-Packard's 82175 paper as a low cost alternative source for use with the PC-324. It was Nick Pietras who was the first to bring that to my attention. But now I find that some, but not all, of my printouts with the HP paper have faded severely with time. Is anyone else having that problem?

TABLE OF CONTENTS

LINEAR EQUATION SOLUTION INADEQUACY 2
TI-95 SBR WITH HEXADECIMAL ADDRESSES, R. Prins 2
3D ON THE FX-7000G, G. Friel 3
CIRCULAR 7 NECKLACE PERMUTATIONS ON THE TI-95, P. Messer 4
MEMBER LISTING 7, 13
THE TI-81 10
DIAGNOSTIC FOR THE TI-59 13
FLATNESS ON THE TI-74 14
BEST FIT REGRESSION ON THE TI-81, P. Hanson 16
ELLIPSE THROUGH 5 POINTS ON THE TI-74, C. Rabe, P. Hanson 17
MORE ON PROBLEMS WITH ASM & INV ASM, S. Garver, R. Prins 20
10 POINT GAUSS QUADRATURE INTEGRATION ON THE TI-59, R. Prins 22
FUNCTION ANALYSIS PROGRAM FOR THE TI-95, R. Skarda 23
LOOKING BACK, P. Hanson 24
TI-74 MATH MODULE ADDENDUMS 25
ELLIPSE THROUGH 5 POINTS ON THE TI-81, P. Hanson 26
HARDWARE AVAILABILITY 23
PRINTED MATTER AVAILABILITY 12, 13, 21, 28

We continue to offer magnetic tapes for TI-74 and TI-95 programs from the newsletter. Send five dollars for a tape with ten programs.

We also continue to offer magnetric card service for TI-59 programs from the newsletter. One dollar per card plus a SASE.

PCH